

risetset_notebook

March 29, 2023

1 risetset Examples

This notebook demonstrates the J `risetset` script. `risetset` computes the rise, transit, and set times of named [IAU Stars](#).

To run this notebook you must install a J jupyter kernel. See Martin Saurer's [GitHub repository](#) for instructions.

```
[1]: NB. J version and date
smoutput 9!:14 ''
smoutput 6!:0 ''

NB. set portable box drawing characters
portchars=:[: 9!:7 '+++++++|-'_ [ ]
portchars 0
```

```
j9.4.1/j64avx512/windows/commercial/www.jsoftware.com/2023-02-27T15:21:53/clang-
15-0-7/SLEEF=1
2023 3 29 22 20 41.553
```

1.1 Installation

`risetset` is distributed as a J addon. It is installed in the J `~addons/jacks` folder. It can be installed from [GitHub](#) with:

```
[2]: load 'pacman'
NB. smoutput install 'github:bakerjd99/jackshacks' NB. uncomment
```

The `jacks` (J-hacks) addons are self contained [JOD generated](#) J scripts. Each `ijs` script is accompanied with pdf document that describes how to use it. Some scripts, like `risetset` are also packaged with a Jupyter notebook (this file) and a pdf version of the notebook. For example, the `risetset` files are:

```
risetset.ijs
risetset.pdf
risetset_notebook.ipynb
risetset_notebook.pdf
```

There are other scripts in `~addons/jacks` and more will be added from time to time. To refresh the folder, reissue the install command.

In addition to these files the subfolder `~addons/jacks/testdata` contains data files. `riserset` files in `testdata` are:

```
Bright_Stars_Meridian_Almanac_23mar27.md
iau_named_stars_2022.txt
Navigation_Stars.txt
```

```
[3]: NB. addon files
dir '~addons/jacks'
```

```
testdata          <dir>      28-Mar-23 22:45:21
brandxmp.ijs       13295 28-Mar-23 22:45:21
brandxmp.pdf       125252 28-Mar-23 22:45:21
gpxutils.ijs       17079 28-Mar-23 22:45:21
gpxutils.pdf       134365 28-Mar-23 22:45:21
ipynb.ijs          4699 28-Mar-23 22:45:21
ipynb.pdf          86966 28-Mar-23 22:45:21
manifest.ijs       1214 29-Mar-23 13:18:17
riserset.ijs       34195 29-Mar-23 13:49:12
riserset.pdf       174094 29-Mar-23 13:18:17
riserset_notebook.ipynb 6671 29-Mar-23 13:52:41
riserset_notebook.pdf 35069 29-Mar-23 13:52:19
```

1.2 Using riserset

```
[4]: NB. load riserset
load '~addons/jacks/riserset.ijs'

NB. version
smoutput 'NB. vmd: ' , , '0,p<; >q<; >0,0' (8!:2) VMDriserset
```

```
NB. (riserset) interface word(s): 20230329j134912
NB. -----
NB. iau_today  NB. named IAU stars rising/setting today
NB. loadstars  NB. loads riserset star data
NB. riserset   NB. rise, transit, set times of stars
```

```
NB. vmd: 0.8.5; 4; 29 Mar 2023 13:49:12
```

```
[5]: NB. set a location - add your own by cloning and modifying location verbs
location_yellowstone
```

```
3 : 0
```

```
NB.*location_yellowstone v-- set parameters for Old Faithful location.
```

```

NB.
NB. monad:  location_yellowstone uuIgnore
NB.
NB.  location_yellowstone 0  NB. set location
NB.  iau_today 0             NB. uses set location with current date
NB.
NB.  NB. uses location with set date
NB.  (location_yellowstone 0) iau_today 0

ymd=. 2013 5 7  NB. mom

NB. longitude, latitude with standard signs
OBSLOCATION_riseset_=: _110.82792 44.46057

UTCOFFSET_riseset_=: 6.0  NB. MST time zone
LIMITMAG_riseset_=: 6.0  NB. stellar magnitude
LIMITHORZ_riseset_=: 10  NB. degrees above horizon

ymd;OBSLOCATION;UTCOFFSET;LIMITMAG;LIMITHORZ
)

```

```
[6]: location_yellowstone 0
```

```

NB. star name, transit altitude, transit time hours, minutes
IAU=: iau_today 0
smoutput #IAU
smoutput 5 {. IAU

```

```

243
+-----+-----+
|Chertan      |61 0 13 |
+-----+-----+
|Zosma        |66 0 13 |
+-----+-----+
|Alula Australis|77 0 17 |
+-----+-----+
|Alula Borealis |78.5 0 17|
+-----+-----+
|Denebola      |60 0 48 |
+-----+-----+

```

Detailed rise and set information is provided by `riseset`.

```

[7]: LB=: _116.375956 43.646775  NB. Meridian
YMD=: 2023 3 27
UO=: 6  NB. MST UTC offset

```

NB. star name, (0=rises/sets), altitudes, times fractional day, times hours, minutes

```
smoutput Rs=: (YMD;U0;LB) riseset 'Algol';'Rigel';'Spica'
```

```
+-----+-----+
|Algol|0| 0.5  0.2910386461449466  6 59|
|      | |87.5  0.6908345293917257 16 35|
|      | | 0.5  0.09337060171945516  2 14|
+-----+-----+
|Rigel|0|  1    0.5527905462948185 13 16|
|      | | 38    0.7780712498266437 18 40|
|      | |0.5  0.006088476753742347  0  9|
+-----+-----+
|Spica|0|  1    0.9012706582406574 21 38 |
|      | | 35 _0.8793144933635634  2 54 |
|      | |0.5  0.3373585646884121  8  6 |
+-----+-----+
```

The stars listed by riseset come from IAU named stars.

[8]: NB. leading characters from UTF-8 CSV IAU star data file

```
800 {. read jpath '~addons/jacks/testdata/iau_named_stars_2022.txt'
```

```
IAU_Name,Designation,HIP,Bayer_Name,Nm,WDS_J,Vmag,RA_J2000,Dec_J2000,Origin,Source,
ID,Const,Etymology_Note
Absolutno,X0-5,X0-5,_ Lyn,_,_,12.13,116.716506,39.094572,2019 IAU100
NameExoWorlds,https://www.nameexoworlds.iau.org/2019approved-names,_,Lyn,Czech
Republic proposal;Absolutno is a fictional miraculous substance in the sci-fi
novel Továrna na absolutno (T...
Acamar,HR 897,13847, 1 Eri,A,02583-4018,2.88,44.565311,-40.304672,,, 1,Eri,
Achernar,HR 472,7588, Eri,A,-,0.45,24.428523,-57.236753,Arabic,, ,Eri,The name
was originally Arabic: ʾāḥīr an-nahr ('river's end').
Achird,HR 219,3821, Cas,A,00491+5749,3.46,12.276213,57.815187,,, ,Cas,"first
applied to Cassiopeiae in the Skalnate Pleso
```

[9]:

```
loadstars~ 2
smoutput 'Named stars:', ":#IAU_Name
smoutput 10 {. IAU_Name
```

Named stars:449

```
+-----+-----+-----+-----+-----+-----+-----+-----+
|Absolutno|Acamar|Achernar|Achird|Acrab|Acrux|Acubens|Adhafera|Adhara|Adhil|
+-----+-----+-----+-----+-----+-----+-----+-----+
```

Additional stars/objects can be added by editing the IAU file or by doing the following.

New objects need a name, right ascension (RA), and declination (Dec) for the J2000.0 epoch.

```
[10]: NB. meeus pg. 99,100
LB=: _71.0833 42.3333 NB. Boston
YMD=: 1988 3 20
UO=: 0
NB. add objects not in IAU names - needs - name, ra, dec
AOB=. (<:;'Venus'),(<41.73129),<18.44092
AOB=. ,&.> (:;'OBJ_Name OBJ_RA_J2000 OBJ_Dec_J2000') ,. AOB
DeltaTsOverride_riseset_=: 56
Vrs=: (YMD;UO;LB;<AOB) riseset 'Venus'
0 0$erase 'DeltaTsOverride_riseset_'
smoutput Vrs
```

```
+-----+-----+
|Venus|0| 1 0.5211284270665463 12 30|
|      | | 66 0.8169433896164773 19 36|
|      | | 0.5 0.1154978057116963 2 46|
+-----+-----+
```

1.3 Maintaining and modifying riseset

All **riseset** code, documentation and test scripts are stored in the JOD dictionary **futs**. To change the code or run the test cases you need to install the JOD dictionaries **futs** and **utils**.

Use J's package manager to install the JOD addons **general/jod**, **general/joddocument**. If you have installed all the addons JOD is already on your system.

After installing JOD do:

1. Download the JOD dump scripts:

<https://github.com/bakerjd99/joddumps/blob/master/utils.ijs>

<https://github.com/bakerjd99/joddumps/blob/master/futs.ijs>

and put them in a **~temp** folder.

2. Start JOD and check for the presence of **futs** and **utils**.

```
NB. start JOD
load 'general/jod'
(;;'futs utils') e. od''
```

3. Only if both dictionaries are missing do:

```
newd 'utils' NB. creates utils dictionary in '~user/joddicts/utils'
newd 'futs' NB. creates futs in '~user/joddicts/futs'
```

4. Load the dictionaries:

```
NB. load utils first
od 'utils' [ 3 od ''
0!:0 <jpath '~temp/utils.ijs'
NB. rebuild references
0 globs&> }. revo ''
```

```

NB. take first binary backup
packd 'utils'

NB. load futs with utils on path
od ;:'futs utils' [ 3 od ''
0!:0 <jpath '~temp/futs.ijs'
NB. rebuild references
0 globs&> }. revo ''
NB. take first binary backup
packd 'futs'

NB. close dictionaries
3 od ''

```

The rest of this notebook *assumes* you have installed **futs** and **utils**.

It also assumes a basic knowledge of JOD. See the JOD Manual for details. The JOD Manual is distributed in the **general/joddocument** addon - see:

```
~addons/general/joddocument/pdfdoc/jod.pdf
```

jod.pdf is also available on [The JOD Page](#)

1.4 riseset test suite

Many **riseset** test cases are in **futs**. Groups of test cases are called suites. The contents of the **riseset** suite is:

```

[11]: NB. open futs and utils - assumed open until notebook end
load 'general/jod'
od ;:'futs utils' [ 3 od ''

```

```

++-----+-----+
|1|opened (rw/ro) ->|futs|utils|
++-----+-----+

```

```

[12]: NB. list test cases in (riseset) suite
, . 3 grp 'riseset'

```

```

+-----+
|1      |
+-----+
|riseset_atan2_smoke |
+-----+
|riseset_espenak_smoke|
+-----+
|riseset_meeus_smoke |
+-----+
|riseset_riseset_smoke|
+-----+

```

```
|risetset_tanner_smoke |
+-----+
```

```
[13]: NB. show test case
1 disp 'risetset_risetset_smoke'
```

```
NB.*risetset_risetset_smoke t-- (risetset) smoke tests.
NB.
NB. created: 2023mar27
NB. changes: -----
```

```
load 'risetset'
```

```
NB. meeus pg. 99,100
LB=: _71.0833 42.3333 NB. Boston
YMD=: 1988 3 20
UO=: 0
NB. add objects not in IAU names - needs - name, ra, dec
AOB=. (<: 'Venus'),(<41.73129),<18.44092
AOB=. ,&.> (: 'OBJ_Name OBJ_RA_J2000 OBJ_Dec_J2000') ,. AOB
DeltaTsOverride_risetset_=: 56
Vrs=: (YMD;UO;LB;<AOB) risetset 'Venus'
0 0$erase 'DeltaTsOverride_risetset_'
```

```
NB. values are within 10 minutes of the meeus book
NB. result - not great but good enough for demo work
Meeusmin=: +/" 1 ] 60 1 *"1 ] 12 25 , 19 41 ,: 2 55
10 > >./|Meeusmin - +/" 1 ] 60 1 *"1 ] _2 {."1 ;2 {"1 Vrs
```

```
LB=: _116.375956 43.646775 NB. Meridian
YMD=: 2023 3 27
UO=: 6 NB. MST UTC offset
```

```
Rs=: (YMD;UO;LB) risetset 'Algol'
Rs=: (YMD;UO;LB) risetset 'Algol';'Rigel';'Spica'
```

```
NB. Bright Stars for 2023 3 27 Meridian
NB. https://www.almanac.com/astronomy/bright-stars/zipcode/83646/2023-03-27
Bs=: ;:'Altair Deneb Fomalhaut Algol Aldebaran Rigel Capella Bellatrix'
Bs=: Bs,;:'Betelgeuse Sirius Procyon Pollux Regulus Spica Arcturus Antares Vega'
```

```
Rs=: (YMD;UO;LB) risetset Bs
```

```
NB. transits match fairly well rise/sets differ 5 to 10 minutes
BsTransit=: 9 18,10 8,12 25,16 35,18 2,18 41,18 43,:18 51
BsTransit=: BsTransit , 19 21,20 11,21 5,21 11,23 34,2 54,3 44,5 58,:8 4
```

```
NB. transit altitude degrees
```

```

BsAlt=: 55 88 16 87 62 38 87 52
BsAlt=: BsAlt,53 29 51 74 58 35 65 19 85

TMP=: {: "1 Rs
ALT=: ((<1;,0)&{&> TMP) ,. BsAlt
TRT=: ((<1;2 3)&{&> TMP) ,. BsTransit

NB. altitudes match to 1 degree
1 = >./ -/"1 ALT

NB. transit times match to 1 minute in worst case
1 = >./ | (60 #:~:~1 ] 0 1 {"1 TRT) - 60 #:~:~1 ] 2 3 {"1 TRT

'IAU NAV'=: loadstars 0
({."1 NAV)=: {: "1 NAV
({."1 IAU)=: {: "1 IAU
Navrs=: (YMD;UO;LB) riseset Nav_Star_Name
Iaurs=: (YMD;UO;LB) riseset IAU_Name

NB. default
Meridianrs=: iau_today 0

NB. date of Uluru star party diner
uYMD=: 2022 10 19
ULURU=: 131.01941 _25.34301
uUTC=: _9.5
uLMAG=: 6.0
uLHORZ=: 5
Ulururs=: (uYMD;ULURU;uUTC;uLMAG;uLHORZ) iau_today 0

0 0$erase 'AOB Meeusmin Vrs LB YMD UO Rs Bs BsTransit BsAlt TMP ALT TRT Navrs
Iaurs'
0 0$erase (;:'IAU NAV') , ({."1 NAV), {"1 IAU
0 0$erase 'uYMD ULURU uUTC uLMAG uLHORZ Meridianrs Ulururs'

smoutput 'PASSED:: riseset_riseset_smoke'

```

[14]: *NB. run all the test cases in the suite*
NB. suppressing all but (smoutput) output
NB. Each test will show PASSED:: if OK.
4 rtt 'riseset'

```

NB. (riseset) interface word(s): 20230329j134912
NB. -----
NB. iau_today NB. named IAU stars rising/setting today
NB. loadstars NB. loads riseset star data
NB. riseset NB. rise, transit, set times of stars

```



```

PASSED:: riseset_atan2_smoke
NB. (riseset) interface word(s): 20230329j134912
NB. -----
NB. iau_today NB. named IAU stars rising/setting today
NB. loadstars NB. loads riseset star data
NB. riseset NB. rise, transit, set times of stars

```

```

PASSED:: riseset_espenak_smoke
NB. (riseset) interface word(s): 20230329j134912
NB. -----
NB. iau_today NB. named IAU stars rising/setting today
NB. loadstars NB. loads riseset star data
NB. riseset NB. rise, transit, set times of stars

```

```

PASSED:: riseset_meeus_smoke
NB. (riseset) interface word(s): 20230329j134912
NB. -----
NB. iau_today NB. named IAU stars rising/setting today
NB. loadstars NB. loads riseset star data
NB. riseset NB. rise, transit, set times of stars

```

```

PASSED:: riseset_riseset_smoke
NB. (riseset) interface word(s): 20230329j134912
NB. -----
NB. iau_today NB. named IAU stars rising/setting today
NB. loadstars NB. loads riseset star data
NB. riseset NB. rise, transit, set times of stars

```

```

PASSED:: riseset_tanner_smoke
1

```

1.5 Building riseset

There are a number of test scripts in `futs` that build and distribute `riseset`. These scripts are tuned to my environment but they do illustrate how to *make* a distribution script.

```

[15]: NB. show main riseset maker
smoutput 1 disp 'build_riseset'
3 od ''

```

```

NB.*build_riseset t-- build (riseset) and distribute.
NB.
NB. created: 2023mar09
NB. changes: -----

coclass tmploc_AAAbuild999_=: 'AAAbuild999' [ coerase <'AAAbuild999'
coinset 'ijod'

```

```

scrn=: 'riseset'

>0{OPENDIC=: did 0

NB. if (imex) is first dictionary on path include it
headdic=: ('imex'-:>1{OPENDIC)#'imex '

>0{od ;: headdic,'futs utils' [ 3 od ''

>0{tmploc get ;:'gettxt getmd read write showpass sha256 afterstr beforestr jnow
jnowpost timestamp VMD',scrn,' ',scrn,'_hashdateurl'

NB. insert/replace build time stamp on post processor
>0{'rc ncv'=: MACRO_ajod_ get 'POST_',scrn
>0{MACRO_ajod_ put (<jnowpost ;2 { ncv) 2} ncv=: ,ncv

NB. update VMD
vmd=: ".'VMD',scrn
builddtm=: timestamp ''
('VMD',scrn)=: (0{vmd),(<1+>1{vmd),<builddtm
>0{tmploc put 'VMD',scrn

smoutput 'building version -> ';".'VMD',scrn

NB. get history document
NB. (histmd,'_md')=: MACRO_ajod_ disp (histmd=: 'HISTORY_',scrn),'_md'

NB. get todo document
NB. (todomd,'_md')=: MACRO_ajod_ disp (todomd=: 'TODO_',scrn),'_md'

NB. generate load script
rc [ 'rc msg file'=: mls scrn

NB. generate companion - contains words used to test main group
rc [ 'rc msg2 file2'=: mls scrn,'Utils'

NB. update hash
ghash=: ".'scrn','_hashdateurl'
(scrn,'_hashdateurl')=: ((sha256 read file);builddtm) (0 1)} ghash
>0{tmploc put scrn,'_hashdateurl'

NB. update distribution files
(3 : 0) file
if. IFWIN do.
    smoutput 'saved in ~JACKSHACKS'

    djacksd=. jpath '~JACKSHACKS/'

```

```

scr=. djacksd,scrn,'.ijs'
(toHOST file) write scr

NB. copy associated files
'Dname Dobj'=. ({. , {:) 0{DPATH__ST__JODobj
if. Dname -: 'futs' do.
  NB. local (futs) document directory
  docd=. ". 'DOC_',Dobj,'_'
  (read docd,'riseset.pdf') write djacksd,'riseset.pdf'
end.
jtmpd=. jpath '~temp\'
(read jtmpd,'riseset_notebook.ipynb') write djacksd,'riseset_notebook.ipynb'
(read jtmpd,'riseset_notebook.pdf') write djacksd,'riseset_notebook.pdf'

dtestd=. jpath '~JACKSHACKS/testdata/'
dtestd gettxt 'iau_named_stars_2022_txt'
dtestd gettxt 'Navigation_Stars_txt'
dtestd getmd 'Bright_Stars_Meridian_Almanac_23mar27_md'
elseif. IFUNIX do.
  smoutput scr=. 'NIMP: save in ~JACKSHACKS'
  NB. scr=. jpath '~/git/jackshacks/',scrn,'.ijs'
  NB. (toHOST file) write scr
elseif.do.
  1 [ showpass 'distributed on WIN/UNIX only' return.
end.

if. fexist scr do.
  1 [ (read y) write showpass scr
else.
  1 [ showpass 'missing distribution script ->';scr
end.
)

cocurrent tmploc_AAAbuild999_
>0{od }. OPENDIC [ 3 od ''

cocurrent 'base'
coerase <tmploc_AAAbuild999_

+-+-----+-----+
|1|closed ->|futs|utils|
+-+-----+-----+

```

1.6 All done - thanks for playing

[]: